
Since the introduction of the Microsoft FoxPro database management system for Windows, developers have been able to use OLE technology to embed or link documents from other applications directly into FoxPro tables. Now Visual FoxPro 3.0 goes much further, allowing you to use OLE Automation to add the functionality in Microsoft Office products—such as Microsoft Word and Microsoft Excel—to applications created with FoxPro.

An OLE Automation server application allows other, external applications to control it by exposing certain parts of its internal structure in the form of objects. These objects have properties and methods. Properties allow you to set or retrieve values that might represent the state of the object. Methods allow you to perform an action on a specific object, such as opening a new document, printing a file, or recalculating a range of cells.

Quick review
You need to consider only two FoxPro functions to retrieve another application's objects: CreateObject and GetObject. Once you have obtained the object, you can immediately begin to access its properties and methods.

CreateObject creates an object from a class definition or an OLE object. It is also used to create objects of built-in FoxPro or user-defined classes:

<memvar> = CREATEOBJECT(cClassName)
cClassName specifies the OLE object from which the new object is created. FoxPro will search for the class name you specify in the following order:

1. Visual FoxPro base classes.

2. User-defined class definitions in memory in the order they were loaded.

3. Classes in the current program.

4. Classes in .VCX class libraries opened with SET CLASSLIB.

5. Classes in procedure files opened with SET PROCEDURE.

6. Classes in the Visual FoxPro program execution chain.

7. The Windows registry (for OLE objects).

OLE objects are created using the following syntax for cClassName:

ApplicationName.Class
For example, to create a Microsoft Excel object using OLE Automation, use the following:

xl = CREATEOBJECT("Excel.Application")
When this code is run, Microsoft Excel is started but is not visible. If Microsoft Excel is already running, another separate instance is created when you execute this command.

GetObject activates an OLE Automation object and creates a reference to the object. For example:

GETOBJECT([cFileName [, cClassName]])
cFileName specifies the full path and filename to activate. The application does not need to be specified, because the OLE dynamic-link libraries (DLLs) determine which application to start based on the filename provided.

For example, the following code launches Microsoft Excel, opens a file named BUDGET.XLS, and creates a reference through an object memory variable named xlBudget:

xlBudget = GETOBJECT("C:\EXCEL\WORK\BUDGET.XLS")
cClassName specifies the class name of the object to retrieve. Some applications can store more than one object type in the same file, allowing you to use the class name to specify the object to activate. For example, if a word-processing application stores its documents, macro definitions, and toolbar objects in the same file, you can create a reference to the document file with the following command:

mDocFile = GETOBJECT("C:\WRDPROC\MYDOC.DOC", "WrdProc.Document")
With some server applications, each time a GetObject function is called an additional instance of the application is started, which uses additional memory. For example, if Microsoft Excel is already running, you can prevent additional instances from starting by omitting cFileName and including cClassName, as in this example:

xl = GETOBJECT(, "Excel.Application")
If Microsoft Excel is not running, a trappable error will be generated. Trapping the error and resorting to the CreateObject function is a common technique to test if an application is already running. An alternative technique is to use dynamic data exchange (DDE) to establish a link with the server application on the System topic, which all DDE applications support. If the link is successfully established, the application is running.

The registration database
The registration database is where FoxPro looks for the class name you specify with the CreateObject and GetObject functions. Under the Windows 3.11 and Windows for Workgroups 3.11 operating systems, entries into the registration database are stored as keys, with a corresponding value for each key in a tree-like structure. You can view the registration database with a program called REGEDIT.EXE, which is located in the directory where you installed Windows. Run REGEDIT with the /V option to see more detail.

Under the Windows NT and Windows 95 operating systems, look under the HKEY_CLASSES_ROOT category or window to find entries for OLE Automation objects. The Windows NT 3.5 registration database editor was invoked by running the REGEDT32.EXE program with no options, as illustrated in the following figure.


Windows NT registration database
Object-oriented extensions in Visual FoxPro 3.0
One of the greatest advantages of using OLE Automation from within Visual FoxPro is the new object-oriented extensions. Because you now can create custom classes, and because classes serve to encapsulate both data and behavior, you can build classes that shield you from the peculiarities of a specific type of OLE server application.

To illustrate this, let's look at the differences between controlling Word and Microsoft Excel. Begin by developing an abstract class that defines how to control all OLE applications. A subclass of this class can then be created for each specific OLE application that you want to control.

#INCLUDE "FOXPRO.H"
 

DEFINE CLASS OLEApplication AS Custom
PROTECTED oOLEApp, ;
cOLERegName, ;
lCloseAppWhenDone
 

oOLEApp = ""
cOLERegName = ""
lCloseAppWhenDone = .T.
 

*— Methods
FUNCTION Init()
IF EMPTY(this.cOLERegName)
=MESSAGEBOX("Cannot create object 
directly from class 
OLEApplication", ;
MB_ICONSTOP, ;
"")
RETURN .F.
ENDIF
 

IF this.AppRunning()
this.oOLEApp = this.GetCurrentInstance()
ELSE
this.oOLEApp = this.CreateNewInstance()
ENDIF
ENDFUNC
 

*— Additional methods get inserted here
ENDDEFINE
This defines the core functionality of your abstract OLEApplication class. First check to ensure that a user of the class is not trying to instantiate it directly. If so, just cancel the INIT method, which prevents the object from being created. Then check to see if the application is running. If it is, create a reference to it. If not, create a new instance.

You need to determine if the application you're trying to control is currently running because certain applications, such as Microsoft Excel, let you create multiple instances, while others, such as Word, don't. Here are two suggestions to determine if an application is already running.

The first involves using the GetObject function to attempt to retrieve a reference to an OLE Automation server. At the time of this writing, this option works fine with Microsoft Excel, but not with Word. Override this function when you define a Word for Windows class, but for now, define this option as a protected function within your OLEApplication class. (You define the function as protected because you will only be calling this function from within the OLEApplication class or any class that you derive from it. There is no need to expose it to the "outside world.")

PROTECTED FUNCTION AppRunning()
*— Returns .T. if app is already running
LOCAL lcOldError, ;
llRunning
llRunning = .T.
lcOldError = ON("ERROR")
ON ERROR llRunning = .F.
*— Attempt to get a reference to a running application
=GETOBJECT(, this.cOLERegName)
ON ERROR &lcOldError
 

this.lCloseAppWhenDone = !llRunning
RETURN llRunning
ENDFUNC
To complete the definition of the OLEApplication class, create two methods, also defined as protected functions, that you will use to create references to new instances or retrieve references to current instances. Notice that each method contains only one statement. You could eliminate the overhead of the function call and place the CreateObject and GetObject calls directly in the INIT method, but then you wouldn't be able to customize the behavior of these methods in your subclasses.

PROTECTED FUNCTION CreateNewInstance()
RETURN CREATEOBJECT(this.cOLERegName)
ENDFUNC
 

PROTECTED FUNCTION GetCurrentInstance()
RETURN GETOBJECT(, this.cOLERegName)
ENDFUNC
The Microsoft Excel class
Define your first subclass to handle OLE Automation with Microsoft Excel. Most of the inherited functionality from the OLEApplication class works just fine with Microsoft Excel. The first thing you need to do is initialize the cOLERegName property with the appropriate name of the Microsoft Excel application object as it appears in the registration database. Then set up a Destroy event method that sends the Quit command to Microsoft Excel whenever the object is being destroyed. The Destroy event method will fire whenever you explicitly release a Microsoft Excel object, or whenever the Microsoft Excel object goes out of scope.

DEFINE CLASS Excel AS OLEApplication
*— Inherited properties
cOLERegName = "Excel.Application"
 

*— Methods
FUNCTION Destroy()
IF TYPE("this.oOLEApp") == "O" AND ;
this.lCloseAppWhenDone
this.oOLEApp.Quit()
ENDIF
ENDFUNC
ENDDEFINE
The Word for Windows class
The second way to determine if an application is already running involves the use of DDE. This is useful in Word because the GetObject function doesn't work properly with Word. To check if an instance is running, it's necessary to use DDE to attempt to establish a link with Word. If you are successful, you know that Word is running, so you immediately terminate the link.

In doing so, you also define a new implementation for the GetCurrentInstance method. This is because of Word's inability to respond to the GetObject function. To get the current instance of Word, you can safely use the GetObject function, passing an empty string as the first parameter.

DEFINE CLASS WinWord AS OLEApplication
*— Inherited properties
cOLERegName = "Word.Basic"
 

PROTECTED FUNCTION AppRunning()
LOCAL lnChannel, ;
llRunning, ;
llDDEOldSafety
 

llDDEOldSafety = DDESETOPTION("Safety")
=DDESETOPTION("Safety", .F.)
lnChannel = DDEINITIATE("WinWord", "System")
IF lnChannel <> -1
*— It's running
this.lCloseAppWhenDone = .F.
=DDETERMINATE(lnChannel)
llRunning = .T.
ENDIF
 

=DDESETOPTION("Safety", llDDEOldSafety)
 

RETURN llRunning
 

PROTECTED FUNCTION GetCurrentInstance()
RETURN GETOBJECT("", this.cOLERegName)
ENDFUNC
 

ENDDEFINE
It is interesting to note that Word does not support creating multiple instances of itself through OLE Automation, while Microsoft Excel does. Using the classes presented here will prevent you from having to worry about inadvertently creating another instance of Microsoft Excel.

Another interesting difference between these two applications is that when releasing a reference to Word that was started through OLE Automation, the Word instance is automatically terminated. However, a reference to Microsoft Excel will not terminate in this fashion. Instead, we must send the Quit method to the Microsoft Excel object, explicitly instructing it to terminate. Furthermore, Microsoft Excel will terminate even if the reference to it was created from a previously running instance.

Providing access to the reference
Note that the reference to the OLE Automation server is a protected member of the OLEApplication class. This means that you can only access this reference through methods of the class or one of its subclasses. There are three approaches you can take to provide access to the reference:

• Unprotect the oOLEApp property. This will provide direct access to the reference from outside the class's methods. This is the most dangerous option, as you then lose control of the reference.

• Create a method that simply returns the reference. This is somewhat less dangerous than the first option, but care must be taken since you now have two references to the same object.

• Create custom methods to manipulate the protected reference. This option is preferred, as you maintain almost complete control over the reference, even though more work must be done initially.

For more details
This article is based on a presentation made by the author at Tech•Ed 95. For a more detailed discussion of this material, please see "Using OLE in Microsoft Visual FoxPro" by Paul Bienick in the July 1995 release of the Microsoft Development Library.

Dealing with Complex Object Models
Microsoft Excel provides more than 100 objects, each of which has a myriad of properties and methods. You don't have to memorize each property and method to be productive with Microsoft Excel but it does help to become familiar with what's there.

A great way to learn how to use these properties and methods is through Visual Basic for Applications, which includes the Microsoft Excel built-in macro language. Visual Basic for Applications uses the same objects, properties, and methods available to FoxPro through OLE Automation. In fact, Visual Basic for Applications itself uses OLE Automation to communicate with Microsoft Excel.

Knowing this, you can use the macro recorder built into Microsoft Excel to record the functionality you are trying to model. The macro recorder generates Visual Basic code, which you can examine to see how various objects are being used. Granted, not every construct in the Visual Basic programming system is directly portable to FoxPro, but Visual Basic interacts with Microsoft Excel in much the same way as FoxPro, and the syntax when dealing with Microsoft Excel objects is almost identical. In many cases you may discover new objects or techniques.

Another great way to become familiar with an application's objects is through the command window, from which you can interactively create objects, set object properties, and call object methods.

